
http://www.mrexcel.com/learn-excel.html 
 
First this week is an interesting question posed by a 
student in one of my classes. She worked in Human 
Resources and wanted to keep a running formula to 
count the number of employees. 
 
Topics 2 & 4 deal with  =RANK and how to deal with 

ties in a Rank. 
 
Topic 3 shows how to take the results of a RANK function and how to virtually sort a list 
using formulas. 

AUTOMATICALLY NUMBER A 
LIST OF EMPLOYEES

Problem: You work in Human Resources. You have a list of employees 
separated by department. As shown in Fig. 369, you have a numerical 
sequence in column A and the employees’ names in column B. Every 
time that you hire or fire an employee, you have to manually renumber 
all of the employees.

Fig. 369

Strategy: Replace the numbers in column A with a formula that will 
count the entries in column B. The formula should count from the cur-
rent row all the way up to row 1.

The COUNT function will not work, because it only counts numeric en-
tries. You need to use the COUNTA function.

• The range that should be counted should extend from B1 to 
the current row. 

• The notation to always use B1 is B$1.

1) As shown in Fig. 370, enter this formula in cell A2: =COUNTA(B$1:
B2).

When you copy this formula down a row, the range that is counted will 
extend from B1 to B3, as shown in Fig. 371. This is because the B2 por-
tion of the above formula is a relative reference that is allowed to change 
as the formula is copied. The dollar sign in the B$1 reference tells Excel 
that no matter where you copy the formula, it should always refer to 
cell 1.

Fig. 370

Fig. 371

2) Copy the formula down to all of the names in your list. They will 
be numbered 1 through 19, just as when you typed the names in 
manually.

When an employee leaves the company, you can simply delete the row, 
All of the other numbers will change to indicate that you now have a 
total of 18 employees, as shown in Fig. 372.

Fig. 372

When you hire a new marketing person, insert a blank row, enter his or 
her name, and then copy any formula from A to the new row.

As shown in Fig. 373, all of the subsequent employees will be 
renumbered.

Summary: While this is a specific example, the concept of using a range 
as an argument where only one portion of the range contains an abso-
lute reference is a common solution to keeping a running total of all cells 
above the current row.

Functions Discussed: =COUNT(); =COUNTA()

RANK SCORES

Problem: You have four writers working on a project. Each week, you 
report how many pages they have written towards their goal, as shown 
in Fig. 374. You wish to have a formula that ranks them in high-to-low 
order.

Fig. 373

Fig. 374

Strategy: If you are not concerned about ties, then use the RANK func-
tion. The function requires two arguments. In plain language, you are 
asking the function to assign a rank to the value in H23 amongst all 
values in H23:H26.

As shown in Fig. 375, in cell F23, use =RANK(H23,$H$23:$H$26). Note 
that the H23:H26 range in the second parameter is in absolute reference 
style due to the dollar signs. This allows the formula to be easily copied 
to each name in the list.

Additional Details: The above version of the function will rank the 
values in high-to-low order. Sometimes you might need to rank in a low-
to-high fashion. Golf is one such instance. You can specify an optional 
third parameter to specify the order. Using a third parameter of 1 will 
force the rank results to be reported in low-to-high order, as shown in 
Fig. 376.

Fig. 375

Fig. 376

Summary: The RANK function is useful for producing a ranking by us-
ing formulas.

Functions Discussed: =RANK()

Cross Reference: Rank a List Without Ties; Sorting with a Formula

SORTING WITH A FORMULA

Problem: In the previous example, you used the RANK function to find 
the relative rank order of four writers, as shown in Fig. 377. Now you 
want to use a formula to produce a sorted list of the writers in high-to-
low sequence.

Strategy: In cells F28 through F31, enter the numbers 1 through 4. Use 
the VLOOKUP function to return the name in column G and the pages 
in column H. Here is how this function works.

VLOOKUP stands for vertical lookup. There are four parameters to the 
VLOOKUP function. In plain language, you are asking Excel to look for 
the value in F28 in the first column of F23:H26. When Excel finds an 
exact matching value, it returns the name in the second column of the 
lookup range.

The first parameter is the value that you are trying to match. In the 
case of cell G28, you would be looking for the value in F28. Write this as 

Fig. 377

$F28 so that you can copy the formula to column H without rewriting 
that parameter.

The second parameter is the database range containing rows and col-
umns of data. The key value that you are looking up must be in the first 
column of the range. In this case, it would be $F$23:$H$26. Note that 
you use dollar signs before both the column letters and row numbers in 
order to keep the database range absolute as you copy the formula.

The third parameter tells Excel the column from which you want to re-
turn the answer. For the name in column G, it is column 2 of the range 
F23:H26. For the page count in column H, it is column 3 of the range 
F23:H26.

The fourth parameter tells Excel if you will allow a close match. If your 
original data is not sorted, you are required to specify an exact match. 
For the fourth parameter, use TRUE for a close match and FALSE for 
an exact match.

Follow these steps:

1) As shown in Fig. 378, enter the following formula in G28: =VLOOK
UP($F28,$F$23:$H$26,2,FALSE).

Fig. 378

2) Copy cell G28 to H28. The result in H28 will also be Josh, as shown 
in Fig. 379.

3) Edit the formula in the formula bar to change the third parameter 
from column 2 to column 3, as shown in Fig. 380.

The result in H28 will now contain the number of pages written by Josh, 
as shown in Fig. 381.

Fig. 379

Fig. 380

Fig. 381

4) Copy G28:H28 down to the next three rows. You will now have a 
sorted list of the data, as shown in Fig. 382.

Additional Details: Your goal is to always enter one formula that you 
can copy to the entire data range. In this case, your formula in G28 could 
be copied to anywhere in column G, but when you copied it to column 
H, the third parameter had to be manually edited. You needed to plan 
ahead to use the proper combination of dollar signs in the references in 
order to ensure that three of the four parameters were correct when you 
copied the formula to column H.

If you find that you have only a few columns of data in an example like 
this, you can edit the third parameter manually. If you have many col-
umns of data, this could get tedious. The =INFO(“Col”,G28) function 
would return a column number of cell G28. This would return a 7 for 
column G and an 8 for column H. You could have used the following 
formula in G28:

=VLOOKUP($F28,$F$23:$H$26,CELL(“Col”,G28)–5,FALSE)

If you enter this formula in G28, you can copy it to all rows and columns 
of your results table.

Summary: After using a RANK function to assign rank values to a 
list, use a second table with the numbers 1 through n and a series of 
VLOOKUP formulas in order to return a sorted list of the data.

Functions Discussed: =VLOOKUP(); =CELL(); =RANK()

Cross Reference: Rank Scores

Fig. 382

RANK A LIST WITHOUT TIES

Problem: The RANK function behaves strangely when there are ties. It 
is possible to have a list where two people are ranked second and no one 
is ranked third. In Fig. 383, Dora and Jerry are ranked second with 90 
units produced. Next is Harry with 86 units. Harry will receive a rank 
of 4.

This behavior is by design. However, if you are later going to use VLOOK-
UPs to sort the employees by productivity, having two people ranked as 
#2 and no one ranked as #3 is not a good situation.

Fig. 383

In the formulas in columns F and G, the spreadsheet designer counted 
on there being one employee at each rank from 1 to 10. Since Excel did 
not assign anyone to a rank of #3 or #7, Jerry and Bill do not show up in 
the list, as shown in Fig. 384.

Strategy: In this case, you absolutely want the list in A4:A13 to be 
ranked without ties. The generally accepted solution may seem rather 
convoluted, but it works. In plain language, the formula in column C 
will say, “Give me the RANK of this value, plus 1 for every row above 
me that has an identical score.” As shown in Fig. 385, this can be accom-
plished with the following formula: 

=RANK(B4,$B$4:$B$13)+COUNTIF(B$3:B3,B4)

Fig. 384

As you copy this formula down, the first parameter of COUNTIF will 
expand to include B3 down to the row above the current row. Thus, in 
cell C13, the formula will be as follows: 

=RANK(B13,$B$4:$B$13)+COUNTIF(B$4:B12,B13)

The COUNTIF portion of the formula counts how many rows above the 
current row have an identical score. For each row above that is a tie, 1 
gets added to the current row. This causes Bill to be ranked seventh in-
stead of sixth. It may not be fair that Ashley appears before Bill, but in 
the summary report, anyone can notice that they have a tie.

Summary: Add a COUNTIF function to the RANK function in order to 
prevent ties.

Functions Discussed: =RANK(); =COUNTIF()

Fig. 385




